

Calculus WIZ a The Mathematical Explorer – použitie

Numerické a symbolické možnosti

Oba softvéry, *Calculus WIZ* a *The Mathematical Explorer* podporujú mnoho symbolických a numerických možností ich materského produktu *Mathematica*. Môže sa využívať aritmetická presnosť výpočtov systému *Mathematica* na vyčíslenie známych matematických konštánt na ľubovoľne veľký počet desatinných miest.

Ludolfovo číslo vyjadrené s presnosťou na 120 desatinných miest:

N[Pi, 120]

```
3.1415926535897932384626433832795028841971693993751058209\
74944592307816406286208998628034825342117067982148086513\
28230665
```

Eulerova konštanta:

N[E, 120]

```
2.7182818284590452353602874713526624977572470936999595749\
66967627724076630353547594571382178525166427427466391932\
00305992
```

S konštantami sa dá pracovať aj symbolicky:

E^(I Pi)

-1

Niekedy systém nedá priamu odpoveď, ako napr. v tomto príklade, kde umocňujeme imaginárnu jednotku na samú seba:

I^I

i^i

Numerická hodnota danej mocniny je

N[I^I]

0.20788 + 0. i

Systém *The Mathematical Explorer* zapíše výsledok aj symbolicky:

ComplexExpand[I^I]

$$e^{-\pi/2}$$

Tento príkaz systému *Mathematica* nie je k dispozícii v programe *Calculus WIZ*. Ak napíšete

ComplexExpand[I^I]

program *Calculus WIZ* dá nasledujúcu odpoveď:

Disabled`ComplexExpand[i^i]

Riešenie algebrickej rovnice je jednoduché

Solve[x^3 == 1, x]

$$\{x \rightarrow 1\}, \{x \rightarrow -\sqrt[3]{-1}\}, \{x \rightarrow (-1)^{2/3}\}$$

a výsledok môžeme aj uložiť do premennej. Symbol % označuje výsledok predchádzajúceho výpočtu a symbol /. zabezpečí substitúciu, t.j., x je nahradené hodnotou uvedenou za znakom šípky (x /. x->a dá a) :

x /. %

$$\{1, -\sqrt[3]{-1}, (-1)^{2/3}\}$$

Výsledné komplexné čísla sa dajú transformovať do algebrických tvarov:

ComplexExpand[%]

$$\left\{1, -\frac{1}{2} - \frac{i\sqrt{3}}{2}, -\frac{1}{2} + \frac{i\sqrt{3}}{2}\right\}$$

a/alebo vyčíslí numericky:

N[%]

$$\{1., -0.5 - 0.866025 i, -0.5 + 0.866025 i\}$$

Ako príklad zložitejšieho symbolického výpočtu uvedieme konverziu trigonometrických a hyperbolických funkcií na exponenciálnu funkciu a naopak.

TrigToExp[Sin[z]]

$$\frac{1}{2} i e^{-iz} - \frac{1}{2} i e^{iz}$$

Ďalší príkaz transformuje komplexnú exponenciálnu funkciu na trigonometrické a hyperbolické funkcie:

ExpToTrig[%]

Sin[z]

Výsledok nie je prekvapujúci. Prekvapujúcou je však nasledujúca aplikácia. Prevedme tretiu mocninu funkcie sínus na exponenciálnu funkciu, upravme pomocou binomickej vety a výsledok prevedme späť na trigonometrické funkcie. Tu je výsledok:

TrigToExp[Sin[z]^3]

$$-\frac{1}{8} i (e^{-iz} - e^{iz})^3$$

Expand[%]

$$\frac{3}{8} i e^{-iz} - \frac{3}{8} i e^{iz} - \frac{1}{8} i e^{-3iz} + \frac{1}{8} i e^{3iz}$$

ExpToTrig[%]

$$\frac{3 \text{Sin}[z]}{4} - \frac{1}{4} \text{Sin}[3z]$$

Ak neveríte, že

$$\frac{3 \text{Sin}[z]}{4} - \frac{1}{4} \text{Sin}[3z] == \text{Sin}[z]^3$$

porovnajte výsledok s matematickými tabuľkami, alebo jednoducho napíšte

$$\text{Simplify}\left[\frac{3 \text{Sin}[z]}{4} - \frac{1}{4} \text{Sin}[3z] == \text{Sin}[z]^3\right]$$

Uvidíte, že je to

True

Funkcia ExpToTrig je podporovaná oboma produktmi a dá sa použiť namiesto funkcie ComplexExpand, ak sa jedná o výrazy typu $-\sqrt[3]{-1}$:

ExpToTrig[$-\sqrt[3]{-1}$]

$$-\frac{1}{2} - \frac{i\sqrt{3}}{2}$$

Programovateľné možnosti Calculus WIZ a The Mathematical Explorer

Ako produkty na báze systému *Mathematica*, oba programy *Calculus WIZ* a *The Mathematical Explorer* podporujú procedurálne, funkcionálne (LISP) programovacie štýly s používaním funkcií a klasické programovacie paradigmy. Z tohto dôvodu nie sú ich kapacity limitované preddefinovanými šablónami a dajú sa použiť tvorivým spôsobom, používatelia si môžu vytvárať svoje vlastné programy.

Jednoducho sa napr. dajú generovať tabuľky funkčných hodnôt:

Table[Sin[x], {x, 0, Pi, Pi/6}]

$$\left\{0, \frac{1}{2}, \frac{\sqrt{3}}{2}, 1, \frac{\sqrt{3}}{2}, \frac{1}{2}, 0\right\}$$


Využitím predností preddefinovaného štýlu vkladania a definovania funkcií môžeme vytvoriť krátke a elegantné programy. Vytvoríme tabuľku derivácií a integrálov danej skupiny funkcií:

({#, D[#, x], Integrate[#, x]} & /@ {Sin[x], Cos[x], Tan[x], Cot[x], ArcTan[x]}) // TableForm

Sin[x]	Cos[x]	-Cos[x]
Cos[x]	-Sin[x]	Sin[x]
Tan[x]	Sec[x] ²	-Log[Cos[x]]
Cot[x]	-Csc[x] ²	Log[Sin[x]]
ArcTan[x]	$\frac{1}{1+x^2}$	$x \text{ArcTan}[x] - \frac{1}{2} \text{Log}[1+x^2]$

Veľmi jednoducho sa dajú tiež nakresliť grafy funkcií a zobraziť v jednom obrázku:

Plot[{Sin[x], Cos[x]}, {x, 0, 2 Pi}]



Programovateľné je aj dosadzovanie, napr. pri riešení rovníc.

```
Sin[x]^2+Cos[x]==1/2 /. Sin[x] -> Sqrt[1-Cos[x]^2]
```

$$1 + \cos[x] - \cos[x]^2 = \frac{1}{2}$$

Symbol % označuje predchádzajúci výsledok:

```
% /. Cos[x] -> z
```

$$1 + z - z^2 = \frac{1}{2}$$

Príkaz Solve vyrieši predchádzajúcu rovnicu, na ktorú sa odvolávame pomocou symbolu %:

```
Solve[%, z]
```

$$\left\{ \left\{ z \rightarrow \frac{1}{2} (1 - \sqrt{3}) \right\}, \left\{ z \rightarrow \frac{1}{2} (1 + \sqrt{3}) \right\} \right\}$$

Výsledky môžeme ihneď použiť na spätné dosadenie:

```
Cos[x] == z /. %
```

$$\left\{ \cos[x] = \frac{1}{2} (1 - \sqrt{3}), \cos[x] = \frac{1}{2} (1 + \sqrt{3}) \right\}$$

a môžeme vyriešiť prvú z daných goniometrických rovníc

```
Solve[%[[1]], x]
```

Solve::ifun Inverse functions are being used by Solve, so some solutions may not be found.

$$\left\{ \left\{ x \rightarrow -\text{ArcCos} \left[\frac{1}{2} (1 - \sqrt{3}) \right] \right\}, \left\{ x \rightarrow \text{ArcCos} \left[\frac{1}{2} (1 - \sqrt{3}) \right] \right\} \right\}$$

Všimnite si, ako systém sám vygeneroval upozornenie, že funkcia nemá inverznú funkciu na danej oblasti. Pretože oba produkty, *Calculus WIZ* aj *The Mathematical Explorer* podporujú výpočty s presnými hodnotami, predchádzajúci výsledok nie je zapísaný ako číslo. Môžeme však požiadať o numerické riešenie:

```
N[%]
```


$$\left\{ \{x \rightarrow -1.94553\}, \{x \rightarrow 1.94553\} \right\}$$

Grafické možnosti

Oba produkty poskytujú mnoho rozličných grafických reprezentácií objektov: základné geometrické tvary (nazývané grafické elementárne útvary - primitívy), 2D a 3D grafické funkcie a vrstevnicová grafika.

Nasledujúci príkaz nakreslí bod a úsečku.


```
Show[Graphics[{Line[{{0, 0}, {1, 1}}], PointSize[0.02],  
Point[{1/2, 1}]}]]
```



Úloha: Nájdite priesečník grafov funkcií $\sin x$ a x^2 v prvom kvadrante.

Nakreslíme grafy funkcií, aby sme získali predstavu, ako riešenie vyzerá:

```
Plot[{Sin[x], x^2}, {x, 0, Pi}]
```



Teraz môžeme hľadať priesečníky výpočtom:

```
Solve[Sin[x] == x^2, x]
```

Solve::tdep : The equations appear to involve the variables to be solved for in an essentially non-algebraic way.

```
Solve[sin(x) == x^2, x]
```

Vo všeobecnosti je pomerne zložitý nájst všeobecné riešenie transcendentnej rovnice v kompaktnej forme. Sem preto limitovaný na hľadanie riešenia pomocou nejakej približnej numerickej metódy. Použijeme preto príkaz FindRoot:

```
{x, Sin[x]} /. FindRoot[Sin[x] == x^2, {x, 0.1}]
```

```
{-2.68703 × 10-8, -2.68703 × 10-8}
```


Voľbou hodnota 0.1 ako začiatkovej hodnoty riešenia x dostávame triviálne riešenie. Z predchádzajúceho obrázka vidíme, že druhé riešenie je niekde medzi 0.5 a 1. Začnime s odhadom hodnoty riešenia 0.5:

```
pt = {x, Sin[x]} /. FindRoot[Sin[x] == x^2, {x, 0.5}]
```

```
{0.876726, 0.768649}
```

Získali sme netriviálne riešenie. Zobrazíme všetko v jednom grafe:

```
Plot[{Sin[x], x^2}, {x, 0, Pi},  
  Epilog -> {PointSize[0.03], Point[{0, 0}], Point[pt]}]
```



Spracovanie dát

Pretože programy Calculus WIZ a The Mathematical Explorer podporujú spracovanie dát, ich používanie nie je obmedzené len na predmet matematika.


V nasledujúcom príklade vytvoríme postupnosť náhodných dát (v reálnych situáciách sa tieto údaje získavajú napr. nejakými meraniami):

```
data = Table[{k, k^2 + 2 k + Random[] / 10}, {k, 1, 10}]
```

Program Calculus WIZ nepodporuje príkaz Random[], preto potrebujeme nejaké externe získané reálne dáta, aby sme mohli uskutočniť nasledujúce výpočty.

```
( 1  3.07191 )  
( 2  8.0146  )  
( 3 15.0765  )  
( 4 24.034   )  
( 5 35.0254  )  
( 6 48.0838  )  
( 7 63.0064  )  
( 8 80.0063  )  
( 9 99.0093  )  
(10 120.059  )
```

```
ListPlot[data, PlotStyle -> {PointSize[0.03]},  
  AxesOrigin -> {0, 0}]
```




Ďalší príkaz zabezpečí vytvorenie lineárnej kombinácie funkcií 1, x a x^2 spĺňajúcu podmienku metódy najmenších štvorcov pre interpoláciu dát:

```
Fit[data, {1, x, x^2}, x]
```

$$1.00074x^2 + 1.9887x + 0.0725026$$

Umiestnime dané dáta a ich interpoláciu do jedného grafu:

```
Plot[1.0007380890493205` x^2 + 1.988699387191992` x +  
0.07250258820562902`, {x, 0, 10},  
Epilog-> {PointSize[0.02], Point/@data}]
```



Napokon je potrebné uviesť, že programy Calculus WIZ a The Mathematical Explorer nepodporujú import a export dát do súborov. Prenos dát sa však dá zabezpečiť kopírovaním cez clipboard.

Záver

Oba produkty, *Calculus WIZ* a *The Mathematical Explorer* sú samostatné softvérové produkty na báze technológie systému *Mathematica*. Sú vhodné na objavovanie veľkých matematických myšlienok pomocou vytvorených šablón, ale aj na používanie prác vytvorených študentmi samostatným programovaním. Oba produkty podporujú vnútorný programovací jazyk systému *Mathematica* (s istými obmedzeniami). Oba produkty tiež podporujú mnoho príkazov systému *Mathematica*, napr. príkazy na úpravu výrazov, na symbolické/numerické integrovanie, atď. Nepodporujú príkazy systému *Mathematica* pre spracovanie veľkých dátových množín, pretože sa nepredpokladá, že by tieto príkazy používali študenti stredných škôl a prvých ročníkov univerzít. Je potrebné spomenúť, že existujú ešte aj iné niektoré iné, nepodstatné obmedzenia oproti plne funkčnej verzii systému *Mathematica*.

Calculus WIZ a *The Mathematical Explorer* sú oba odporúčané pre vyššie ročníky stredných škôl a/alebo pre učiteľov bakalárskych stupňov štúdia na univerzitách a vysokých odborných školách.